
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of Multilayer test case 13.3.1.2 which are part of the LTE test suite.

The test case can be demonstrated to run with two LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Cases:
tc_13_3_1_2
Test Group:
Multilayer
ATS Version:
iwd-EUTRA-TVB2009-12_D10wk42
System Simulator used:
Anite Conformance Toolset
UE(s) used:                       Samsung Tanum , LG L2000
Verification Status:
PASS
4 Corrections required for test case 13.3.1.2
4.1 Introduction

This section describes the changes required to make test case 13.3.1.2 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-12_D10wk42 release.

4.2 Change 1

	Testcase name
	13.3.1.2

	Protocol
	Multilayer

	Function Name
	function f_TC_13_3_1_2_EUTRA ( )

	Reason for change
	1. Uplink grant should be stopped on cell1 prior to sending the data to avoid the race condition where UE loops back data on cell1 before the cell power level change can take effect.
2. PDCP count handover should be done prior to cell1 power level change as PDCP resources are released after cell switch off.
3. Security should be configured on cell 2 prior to transmission of rrcConnectionRestablishment.

4. At step 6, rrcConnectionReconfiguration message should be used to resume existing radio bearers. 
5. At step 7, UE might send loopback data and rrcConnectionReconfigurationComplete almost simultaneously. In case TTCN processes the DRB data first, it may lead to failure of test case. So test case should handle the scenario such that either data or signalling at step 7 and 8 can come one after the other. 
6. Postamble should happen on cell 2


	Summary of change
	1. Stopped uplink grant on cell1 prior to sending data
2. PDCP count handover is done prior to changing power level of cell1
3. AS security configured on cell 2 prior to transmission of rrcConnectionRestablishment. 

4. At step 6, in rrcConnectionReconfiguration existing radio bearers are being resumed. 

5. Alt statement added to handle the scenario of DRB data getting processed earlier than rrcReconfigurationComplete. 

6. cell id changed to cell 2 in postamble. 



	Source of change
	File : …\13\MultiLayer_Procedures.ttcn


Before:
	function f_TC_13_3_1_2_EUTRA()

      runs on EUTRA_PTC {

  /* Intra-system connection re-establishment / Re-establishment of a new connection when further data is to be transferred */

    var template (value) CellPowerList_Type v_CellPowerList_AtT0, v_CellPowerList_AtT1;

    var ShortMAC_I v_ShortMAC_I;

    var PhysCellId v_PhysicalCellIdentity_Cell1;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var NextHopChainingCount v_NextHopChainingCount := 0; // Default configuration value according to TS 36.508 Table 4.6.1-10

    //Init cell array and security

    f_EUTRA_Init ( c1 );

    //Get cell specific parameters for Cell 1

    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId ( eutra_Cell1 );

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_Cell2 );

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated ( eutra_Cell2 );

    //Create and configure cell

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    f_EUTRA_CellConfig_Def ( eutra_Cell2 );

    //The UE is in state Loopback Activated (state 4) on Cell 1 according to [18] using the UE TEST LOOP MODE B.

    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB );

    f_EUTRA_RbEst_Def ( eutra_Cell1 );

    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );

    //Set initial cell power levels

    v_CellPowerList_AtT0 := {

      cs_CellPower ( eutra_Cell1, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_Cell2, tsc_NonSuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT0, cs_TimingInfo_Now );

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell2 );

    f_EUTRA_TestBody_Set ( true, layer2 );

    //@siclog "Step 1" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 1. */

    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,

                                             tsc_RbId_DRB1,

                                             cs_TimingInfo_Now,

                                             { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address,

                                                                             px_IPv4_Address ) } ) );

    //@siclog "Step 2" siclog@

    /* The SS changes the power level of cell 1 to non-suitable "Off" and changes the power level of cell 2 to suitable

               according to TS 36.508 subclause 6.2.2.1 in order that the radio link quality of Cell 1 is degraded. */

    v_CellPowerList_AtT1 := {

      cs_CellPower ( eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE ),

      cs_CellPower ( eutra_Cell2, tsc_SuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT1, cs_TimingInfo_Now );

    //@siclog "Step 3" siclog@

    /* Check: Does the UE initiate an RRC connection re-establisment procedure and transmit an RRCConnectionReestablishmentRequest

               message on Cell 2?  */

    SRB.receive( car_SRB0_RrcPdu_IND( eutra_Cell2,

                                      cr_508_RRCConnectionReestablishmentRequest ( tsc_C_RNTI_Def,

                                                                                   v_PhysicalCellIdentity_Cell1,

                                                                                   cr_ReestablishmentCause_OtherFailure,

                                                                                   v_ShortMAC_I ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 3");

    // Start UL grant transmission in Cell 2

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now );

    //PDCP count shall be kept in PDCP re-establishment for AM DRB

    f_EUTRA_SS_PdcpCount_Handover( eutra_Cell1,

                                   eutra_Cell2,

                                  { tsc_DRB1 } );
    //@siclog "Step 4" siclog@

    /* The SS transmits RRCConnectionReestablishment message. */

    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell2, tsc_RRC_TI_Def, v_NextHopChainingCount );

    // Activate RRC security at SS target cell (Cell2) for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell2, f_EUTRA_Security_Get ( ), v_NextHopChainingCount );

    f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 5" siclog@

    /* UE transmits an RRCConnectionReestablishmentComplete message */

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2,  cr_508_RRCConnectionReestablishmentComplete( tsc_RRC_TI_Def ) ) );

    //@siclog "Step 6" siclog@

    /* The SS transmits RRCConnectionReconfiguration message */

    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_Cell2,

                                     cs_TimingInfo_Now,

                                     cs_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est ( tsc_RRC_TI_Def,

                                                                                        v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                        v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                        v_AntennaInfo ) ) );
    //@siclog "Step 7" siclog@

    /* UE transmits an RRCConnectionConfigurationComplete message */

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    //@siclog "Step 8" siclog@

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context on Cell 2? */

    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell2,

                                                    tsc_RbId_DRB1,

                                                    ?,

                                                    { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address,

                                                                                    px_IPv4_Address ) } ) );
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 8");

    f_EUTRA_TestBody_Set ( false );

    //Postamble

    f_EUTRA_Postamble ( eutra_Cell1, E3_TESTMODE );

  }


After:
	  function f_TC_13_3_1_2_EUTRA()

      runs on EUTRA_PTC {

  /* Intra-system connection re-establishment / Re-establishment of a new connection when further data is to be transferred */

    var template (value) CellPowerList_Type v_CellPowerList_AtT0, v_CellPowerList_AtT1;

    var ShortMAC_I v_ShortMAC_I;

    var PhysCellId v_PhysicalCellIdentity_Cell1;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;

    var EUTRA_SecurityParams_Type v_Auth_Params;

    var NextHopChainingCount v_NextHopChainingCount := 0; // Default configuration value according to TS 36.508 Table 4.6.1-10

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex;

    var SR_ConfigIndex_Type v_SR_ConfigIndex;
    //Init cell array and security

    f_EUTRA_Init ( c1 );

    //Get cell specific parameters for Cell 1

    v_PhysicalCellIdentity_Cell1 := f_EUTRA_CellInfo_GetPhyCellId ( eutra_Cell1 );

    //Get FDD or TDD mode from Cell-configuration

    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info ( eutra_Cell2 );

    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated ( eutra_Cell2 );

    //Create and configure cell

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    f_EUTRA_CellConfig_Def ( eutra_Cell2 );

    //The UE is in state Loopback Activated (state 4) on Cell 1 according to [18] using the UE TEST LOOP MODE B.

    f_EUTRA_Preamble ( eutra_Cell1, STATE2A_TESTLOOP_ModeB );

    f_EUTRA_RbEst_Def ( eutra_Cell1 );

    f_EUTRA_CloseUE_TestLoopModeB ( eutra_Cell1, '00'O );

    //Set initial cell power levels

    v_CellPowerList_AtT0 := {

      cs_CellPower ( eutra_Cell1, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_Cell2, tsc_NonSuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT0, cs_TimingInfo_Now );

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell2 );

    f_EUTRA_TestBody_Set ( true, layer2 );

    // stopped UL grant transmission on cell so that UE doesnt loop back the data 

    // on DRB1 of cell 1. 

    f_EUTRA_StopULGrantTransmission ( eutra_Cell1, cs_TimingInfo_Now );
    //@siclog "Step 1" siclog@

    /* The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 1. */

    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU ( eutra_Cell1,

                                             tsc_RbId_DRB1,

                                             cs_TimingInfo_Now,

                                             { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address,

                                                                             px_IPv4_Address ) } ) );

    //PDCP count handover should be done prior to cell switch off. Otherwise, PDCP resources are 

    //released after cell switch off. 

    //PDCP count shall be kept in PDCP re-establishment for AM DRB

    f_EUTRA_SS_PdcpCount_Handover( eutra_Cell1,

                                   eutra_Cell2,

                                  { tsc_DRB1 } );
    //@siclog "Step 2" siclog@

    /* The SS changes the power level of cell 1 to non-suitable "Off" and changes the power level of cell 2 to suitable

               according to TS 36.508 subclause 6.2.2.1 in order that the radio link quality of Cell 1 is degraded. */

    v_CellPowerList_AtT1 := {

      cs_CellPower ( eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE ),

      cs_CellPower ( eutra_Cell2, tsc_SuitableCellRS_EPRE)

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList_AtT1, cs_TimingInfo_Now );

    //@siclog "Step 3" siclog@

    /* Check: Does the UE initiate an RRC connection re-establisment procedure and transmit an RRCConnectionReestablishmentRequest

               message on Cell 2?  */

    SRB.receive( car_SRB0_RrcPdu_IND( eutra_Cell2,

                                      cr_508_RRCConnectionReestablishmentRequest ( tsc_C_RNTI_Def,

                                                                                   v_PhysicalCellIdentity_Cell1,

                                                                                   cr_ReestablishmentCause_OtherFailure,

                                                                                   v_ShortMAC_I ) ) );

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 3");

    // Start UL grant transmission in Cell 2

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now );

    // Activate RRC security at SS target cell (Cell2) for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell2, f_EUTRA_Security_Get ( ), v_NextHopChainingCount );

    f_EUTRA_Security_Set ( v_Auth_Params );
    //@siclog "Step 4" siclog@

    /* The SS transmits RRCConnectionReestablishment message. */

    f_EUTRA_RRC_RRCConnectionReestablishment_Def ( eutra_Cell2, tsc_RRC_TI_Def, v_NextHopChainingCount );

    //@siclog "Step 5" siclog@

    /* UE transmits an RRCConnectionReestablishmentComplete message */

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2,  cr_508_RRCConnectionReestablishmentComplete( tsc_RRC_TI_Def ) ) );

    //@siclog "Step 6" siclog@

    /* The SS transmits RRCConnectionReconfiguration message */

    v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex ( eutra_Cell2 );

    v_SR_ConfigIndex := f_EUTRA_CellInfo_GetSR_ConfigIndex ( eutra_Cell2 );
    //Send RRCConnectionReconfiguration

    SRB.send ( cas_SRB1_RrcPdu_REQ ( eutra_Cell2,

                                     cs_TimingInfo_Now,

                                     cs_RRCConnReconfig_Resume_SRB2_DRBs ( tsc_RRC_TI_Def,

                                                                           v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                           v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                           v_AntennaInfo,

                                                                           v_Sr_PUCCH_ResourceIndex,

                                                                           v_SR_ConfigIndex) ) );   

    //@siclog "Step 7" siclog@

    /* UE transmits an RRCConnectionConfigurationComplete message */

    //@siclog "Step 8" siclog@

    /* Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context on Cell 2? */

    alt {

      []  SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) )

      {

        DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell2,

                                                    tsc_RbId_DRB1,

                                                    ?,

                                                    { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address,

                                                                                    px_IPv4_Address ) } ) );

      }  

      []    DRB.receive ( car_DRB_COMMON_IND_PDCP_SDUList ( eutra_Cell2,

                                                    tsc_RbId_DRB1,

                                                    ?,

                                                    { crs_IPv4_AnyIcmpEchoReply_1 ( px_IPv4_Address,

                                                                                    px_IPv4_Address ) } ) )

      {

        SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

      }                                                                              

    }
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 13.3.1.2 Step 8");

    f_EUTRA_TestBody_Set ( false );

    //Postamble

    f_EUTRA_Postamble ( eutra_Cell2, E3_TESTMODE );

  }


5 Execution Log Files

5.1 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_13_3_1_2_Samsung_log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

5.2 LG L2000 UE

The LG L2000 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_13_3_1_2_LG_log.html:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6 References

	[1]
	R5s100745: This archive comprises html format execution log file.

